TP Maple 7 | Courbes et surfaces

Nous n'explorerons dans ce chapitre que les commandes graphiques essentielles offertes par Maple. Pour chacune des commandes de tracé, nous avons fait le choix de n'exposer que les options les plus utilisées en mettant notament de côté certains aspects esthétiques plus avancés (comme le choix des polices, etc.).

1. Représentation graphique d'une ligne brisée

Tracé d'une ligne brisée

plot(L, ***options de tracé***)

où L est une liste de points donnés sous la forme d'une liste à deux éléments $[x, y]$, respectivement abscisse et ordonnée du point.

```
> plot([[0,0],[2,6],[3,-2],[3,3],[-3,2],[0,0]],thickness=2,color=black);
```



2. Expression vs. fonction

Maple peut représenter graphiquement des expressions ou des fonctions numériques. La syntaxe différera selon le choix de l'utilisateur.

2.1. Représentation graphique d'une expression

Pour un tracé d'une expression E de la variable x sur l'intervalle $[a, b]$ en bloquant l'ouverture verticale à $[c, d]$, on utilisera la syntaxe suivante [1] :

Tracé d'une expression

plot(E, $x = a..b$, $y = c..d$, ***options de tracé***)

```
> plot(cos(x), x=-2*Pi..2*Pi, thickness=2, color=black);
```



L'utilisateur peut choisir l'épaisseur du tracé au moyen de l'option **thickness**, la couleur grâce à **color**, etc.

1. On peut éventuellment avoir a et c égaux à $-\infty$, c et d égaux à $+\infty$, on notera dans ce cas **-infinity** et **infinity** dans la commande **plot**.

```
> plot(sin(x)/x, x=0..infinity, thickness=2, color=black);
```



Il existe beaucoup d'autres options, en voici quelques unes parmi les plus utiles.

Options de **plot**

- **color = blue, black, green *ou* pink, ...** : définit la couleur de la courbe tracée.
- **thickness = 1, 2, ...** : fixe l'épaisseur du trait.
- **scaling = constrained** : impose un repère orthonormé.
- **axes = framed , boxed, normal et none** : styles des axes.
- **style = line, point, ...** : styles de tracés (par défaut line).
- **discont = true ou false** : choisir true en cas de discontinuité.

En faisant le choix **style=*line***, l'utilisateur force Maple à tracer une ligne brisée de pas très petit et modiafiable (cf. la rubrique d'aide concernant la commande **plot**) approchant le graphe de l'expression [2]. Ce mode de tracé générera des bizarreries dans le cas des fonctions discontinues. Il faudra alors utiliser l'option **discont** pour éviter les problèmes. Examinons l'exemple de $\frac{1}{x+1}$...

2. Le style *point* consiste en un tracé point par point (donc assez grossier si le pas est grand).

```
> plot(1/(x+1), x=-2..2, y=-5..5, color=black, thickness=2);
```



Le saut d'une branche de la courbe à l'autre par Maple est matérialisé par une ligne verticale (on peut d'ailleurs considérer cette dernière comme l'asymptote verticale de l'expression en -1). Voici le moyen de la faire disparaître.

```
> plot(1/(x+1), x=-2..2, y=-5..5, color=black, thickness=2, discont=true);
```



L'utilisateur pourra tracer plusieurs courbes dans la même fenêtre en utilisant la syntaxe suivante.

Tracés simultanés d'expressions

plot($[E_1, \ldots, E_n]$**,** $x = a..b$**,** $y = c..d$**, *options de tracé***)

avec la syntaxe suivante pour les options : **color=[*couleur*$_1$, ..., *couleur*$_n$].**

Traçons par exemple les fonctions de $\mathbb{R}_+$ dans $\mathbb{R}$ définie par $x \mapsto \sqrt{x}$ et $x \mapsto x^2$ sur un même graphique.

```
> plot([sqrt(x),x^2],x=0..3,y=0..3,color=[black,grey], thickness=[1,2],
      scaling=constrained);
```



2.2. Représentation graphique d'une fonction

On adoptera la syntaxe suivante pour le tracé d'une fonction f pour une variation de la variable dans $[a, b]$ et en limitant l'ouverture verticale à l'intervalle $[c, d]$:

Tracé d'une fonction

plot(f**,** $a..b$**,** $c..d$**, *options de tracé*)**

Les options suivent la même syntaxe que dans le cas des expressions, nous n'y reviendrons donc pas. Puisque si f désigne une fonction, le logiciel considère $f(x)$ comme une expression, on dispose d'une deuxième possiblité pour le tracé de la courbe représentative de f : **plot(**$f(x)$**,** $x = a..b$**,** $y = c..d$**)** à condition que x soit une variable libre. Traçons par exemple la courbe représentative du logarithme népérien.

```
> plot(ln, 0.01..3, -3..2, color=black, thickness=2);
```



On peut également tracer deux graphes sur la même figure.

```
> plot([cos,sin], -2*Pi..2*Pi, -1..1, color=[black,grey], thickness=[1,2],
      scaling=constrained);
```



Exercice 1.

Soient pour $\lambda > 0$,

$$x > 0 \mapsto f_\lambda(x) = \frac{x\,\ell\mathrm{n}(x/\lambda)}{x^2+1}$$

et Γ_λ le graphe de λ.

1. Tracer sur un même graphique les courbes Γ_λ pour différentes valeurs de λ.

2. Donner l'équation de la tangente $\mathscr{D}_\lambda$ à la courbe Γ_λ au point d'abscisse $a > 0$.

3. Montrer que les droites $\mathscr{D}_\lambda$ sont concourantes en un point $M(a)$ dont on donnera les coordonnées.

4. Tracer le lieu du point $M(a)$ lorsque a décrit $\mathbb{R}_+^*$.

3. Courbes paramétrées

Comme pour les courbes représentatives de fonctions, l'utilisateur pourra utliser soit des expressions, soit des fonctions. Nous nous contentrons d'exposer des tracés d'expressions dans tout ce qui suit, le lecteur adaptera sans peine les différnets résultats au cas des fonctions.

3.1. En coordonnées cartésiennes

Pour tracer la courbe $t \mapsto (x(t), y(t))$ sur $[a, b]$ en restrteingnat le tracé à la fenêtre $[u, v] \times [l, k]$, le lecteur emploiera la syntaxe suivante :

Tracé d'une courbe $t \mapsto (x(t), y(t))$

plot([$x(t)$**,** $y(t)$**,** $t = a..b$]**,** $u..v$**,** $l..k$ **,*options de tracé*)**

Voici une ligne de commandes permettant de tracer une Astroïde[3].

```
> plot([cos(t)^3,sin(t)^3,t=0..2*Pi],-1.2..1.2, -1.2..1.2, color=black,
  thickness=2);
```



On peut également tracer une famille de courbes sur le même graphique en suivant la syntaxe décrite ci-dessous.

Tracés multiples en cartésiennes

plot([***courbe***$_1$**,** …**,** ***courbe***$_n$]**,** $u..v$**,** $l..k$ **,*options de tracé*)**

où ***courbe***$_i$=[$x_i(t)$**,** $y_i(t)$**,** $t = a_i..b_i$] correspond au tracé de la i-ème courbe.

Traçons alors deux Astroïdes...

3. Donnée par le paramétrage $t \mapsto (\cos^3(t), \sin^3(t))$

```
> plot([[cos(t)^3,sin(t)^3,t=0..2*Pi],[3*cos(t)^3,2*sin(t)^3,t=0..2*Pi]],
  -3.6..3.6, -2.4..2.4, color=[black, grey], thickness=[2,1],
  scaling=constrained);
```



Exercice 2.

Soit Γ courbe paramétrée définie par

$$x(t) = \frac{2t-1}{t^2-1} \qquad y(t) = \frac{t^2+3}{t-1}$$

1. Tracer Γ.

2. Étudier les branches infinies de Γ.

3. Calculer les coordonnées du point double de Γ.

Exercice 3.

Soit, pour tout réel t, $z(t) = 2\cos(t) + i\sin(t)$.

1. Tracer la courbe C définie par z.

2. Soit K l'ensemble des centres des cercles passant par l'origine et tangents à C. Donner une représentation paramétrique de K et tracer K.

3.2. En coordonnées polaires

Pour tracer la courbe polaire $\theta \mapsto \rho(\theta)$ sur $[\theta_1, \theta_2]$ en restrteingnat le tracé à la fenêtre $[u, v] \times [l, k]$, le lecteur emploiera la syntaxe suivante [4] :

4. On pourra aussi adopter la syntaxe suivante **plot**($\rho(\theta)$, $\theta = \theta_1..\theta_2$, **coords=polar)** avec toutefois l'inconvéniant de mal contrôler la fenêtre de tracé.

Tracé d'une courbe polaire $\theta \mapsto \rho(\theta)$

plot([$\rho(\theta)$, θ, $\theta = \theta_1..\theta_2$], $u..v$, $l..k$, **coords=polar**, ***options de tracé***)

Les options de tracé sont identiques à celles du tracé des courbes représentatives. Voici une ligne de commandes permettant de tracer le Limaçon de Pascal d'équation $\rho = 1 + 2\cos(\theta)$.

```
> plot([1+2*cos(theta),theta,theta=0..2*Pi],-0.5..3.2, -2.2..2.2,
  coords=polar, color=black, thickness=2);
```



Tracés multiples en polaires

plot([***courbe***$_1$, ..., ***courbe***$_n$], $u..v$, $l..k$,***options de tracé***)

où ***courbe***$_i$=[$\rho(\theta)$, θ, $t = \theta_i..\phi_i$] correspond au tracé de la i-ème courbe.

Traçons quelques courbes de la famille des Limaçons de Pascal...

```
> plot([[1+2*cos(theta),theta,theta=0..2*Pi], [1+0.8*cos(theta),theta,
  theta=0..2*Pi],[1+1.2*cos(theta),theta,theta=0..2*Pi]], coords=polar,
  color=black, thickness=[3,2,1]);
```



Exercice 4.

Soit Γ la courbe d'équation polaire $\rho(\theta) = 1 + \cos(\theta)$. On utilisera le logiciel pour les calculs intermédiaires et les tracés.

1. Tracer Γ.

2. On note A le point de coordonnées polaires $[\theta_0 = 0, \rho_0 = 2]$. Une droite variable $\mathscr{D}_\theta$ passant par O et faisant un angle θ avec l'axe polaire recoupe Γ en deux points P_θ et Q_θ. Déterminer le lieu du centre de gravité du triangle $AP_\theta Q_\theta$ lorsque $\mathscr{D}_\theta$ varie.

3. Prouver que les tangentes à Γ en P_θ et Q_θ sont orthogonales.

4. Déterminer le lieu de leur point d'intersection lorsque $\mathscr{D}_\theta$ varie.

4. Courbes définies par une équation cartésienne

On trouvera dans la librairie **plots** le moyen de tracer des courbes définies *implicitement* par la donnée d'une équation cartésienne $f(x, y) = 0$.

La commande implicitplot

La commande

implicitplot(f, x=a..b, y=c..d, options)

permet de tracer la portion de la courbe d'équation cartésienne $f(x, y) = 0$ contenue dans la fenêtre $[a, b] \times [c, d]$.

```
> implicitplot(cos(x)+cos(y) = 1, x = -4*Pi .. 4*Pi, y = -4*Pi .. 4*Pi);
```



5. Tracé des surfaces et des courbes gauches

Nous rappelons au lecteur que les courbes de l'espace sont également appelées *courbes gauches*. Nous nous contenterons d'un survol des outils 3D de **MAPLE**.

5.1. Courbes de l'espace

L'utilisateur pourra tracer des courbes gauches au moyen de la commande **spacecurve** de la bibliothèque PLOTS. Il faudra en particulier *charger* celle-ci (par la ligne de commandes **with**(PLOTS) avant que n'apparaisse l'instruction **spacecurve**.

```
> with(plots): spacecurve([t*cos(t),t*sin(t),t],t=0..9*Pi, numpoints=2000,
                          color=black, orientation=[50,70], thickness=2);
```



5.2. Les surfaces

C'est la fonction **plot3d** qui permet le tracé des nappes paramétrées $z = f(x, y)$ dans la fenêtre définie par le rectangle $[a, b] \times [c, d] \times [l, k]$ suivant la syntaxe indiquée ci-dessous :

Tracé d'une nappe cartésienne $z = f(x, y)$

plot3d($f(x, y)$**,** $x = a..b$**,** $y = c..d$**,** $z = k..l$**, *options de tracé*)**

Traçons au moyen de **plot3d** une selle de cheval...

```
> plot3d(x^2-y^2,x=-1..1,y=-1..1,color=grey,orientation=[60,60],
  scaling=constrained);
```



Nous ne détaillerons pas ici les options de tracé (comme ***orientation***, permettant d'obtenir de *voir* la surface sous l'angle choisi par l'utilisateur). Le lecteur est renvoyé à l'aide en ligne du logiciel pour de plus amples informations.

La commande **plot3d** permet également le tracé des surfaces paramétrées dans un des sytèmes de coordonnées suivants : cartésien, cylindrique, sphérique, etc.

Tracé d'une nappe paramétrée

plot3d($[x(u,v)$**,** $y(u,v)$**,** $z(u,v)]$**,** $u = a..b$**,** $v = c..d$**, coords=... ,*options de tracé*)**

Notons que l'option ***coords*** désigne le système de coordonnées choisi et vaut par défaut le système cartésien. Traçons par exemple un hyperboloïde à une nappe (H_1 pour les intimes) à partir de

Parmi les options

```
> plot3d([cosh(phi)*cos(theta),cosh(phi)*sin(theta),sin(phi)],
  phi=-arcsinh(1)..arcsinh(1),theta=0..2*Pi,color=grey);
```



5.3. Surfaces définies par une équation cartésienne

On trouvera dans la librairie **plots** le moyen de tracer des courbes définies *implicitement* par la donnée d'une équation cartésienne $f(x, y) = 0$.

La commande **implicitplot3d**

La commande

implicitplot3d(f, x=a..b, y=a'..b',z=a"..b", options)

permet de tracer la portion de la surfae d'équation cartésienne $f(x, y, z) = 0$ contenue dans la fenêtre $[a, b] \times [a', b'] \times [a", b"]$.

```
> with(plots);
  implicitplot3d(x^2+y^2-z^2=0,x=-4..4,y=-4..4,z=-4..4,color=grey,
  scaling=constrained,grid=[15,15,15],orientation=[60, 60])
```



Exercice 5.

L'espace est muni d'un repère orthonormé direct $\mathscr{R}$. Soient $p > 0$, $F(p/2, 0, 0)$ et $\mathscr{P} : x = p/2$.

1. Déterminer une équation cartésienne de l'ensemble $\mathscr{Q}$ des points qui sont centre d'une sphère tangente au plan $\mathscr{P}$ et passant par F. Reconnaître $\mathscr{Q}$.

2. Tracer $\mathscr{Q}$ au moyen de Maple.

3. Comment retrouver la nature de $\mathscr{P}$ en utilisant les propriétés géométriques des paraboles ?